
Reverberation

• Reverberation is a naturally occurring phenomenon where sound is 
reflected across walls and surfaces, leading to a persistence of sound 
after it has been produced.
• As sound is reflected across surfaces, sound waves are absorbed and 

generally reflected at smaller amplitudes.  

• Nearly all sound we hear has reverberance.
• Anechoic chambers are places where no reflections are created.

• Sound generally emanates in all directions.
• Reverb is one of the most common effects in the digital audio world 

and is difficult to replicate naturally



Reverberation

Consider a direct sound source that is an 
impulse (i.e., a single sample of sound) in 
some reverberant space.  The chart on the 
left shows the amplitude of reflections.  
Note how the early reflections are spread 
out and the late reflections become denser 
and decay in amplitude.

This chart depicts an impulse response, 
which details the reflections of a 
room/space based on a single impulse.  
Since sound is a sum of impulses, all 
moments of sound will respond in similar 
fashion.



Key Ideas in Reverberation

• Reverberation time – the time it takes for the reflections to reach -
60db relative to the direct sound source.
• Early reflections that arrive within 50-80 milliseconds are not heard as 

separate from the direct sound.  
• When the early reflections are greater than 50-80ms, we perceive the 

reflections as an echo.  This is why speaking in a canyon where the sound 
travels a great distance before it is reflected creates an echo.

• Pre-delay is the time between the direct sound source and the first 
early reflection.
• Good indicator for the size of the room



Key Ideas in Reverberation

• The absorption of reflections against surfaces is frequency 
dependent.
• Higher frequencies tend to be absorbed more than lower frequencies.
• Lower frequencies tend to have a longer reverberation time.
• Artificial reverberation tends to use some sort of damping of high frequencies 

by passing the final result through a low-pass filter.

• Density of early reflections can suggest the size of the room.
• We should not hear individual echoes.  The echo density should 

ideally be at least one thousand per second.



A Crude Reverb
• Consider the simple feedback comb filter 

below that we examined earlier
• It’s decaying and infinite response mirrors the 

kind of impulse response similar in 
reverberation.
• How can we use the feedback comb filter to 

construct a reverb?

In Parallel

• If we place the comb filters in parallel, how 
many filters would we need to achieve a 
good enough echo density?
• At least about 40!
• The delay for each one should be unique and 

randomized to avoid accentuating certain 
frequencies.  This will give us a mostly flat 
amplitude response, which is a plus!
• Remember individual feedback comb filters 

accentuate certain frequencies.  They do not have a 
flat amplitude response. 

• Efficient?  Nope.  We are using a lot of comb 
filters!

Comb 1

Comb 2

Comb N

Input

Output



(
SynthDef(\parallelCombReverbStereo, {

arg in, out = 0, maxdelaytime = 0.2, decaytime = 1, inputGain = 0.5, reverbGain = 0.2;
var reverb, input, delayTimes, numOfCombs = 40;
input = In.ar(in, 2);
reverb = 0;
numOfCombs.do({

arg delayTime;
reverb = reverb + CombN.ar(

input,
maxdelaytime,
Rand(0, 0.02).asStream.next, // Randomized delay time
decaytime,
1/numOfCombs // Amplitude

)
});
Out.ar(out, (input * inputGain) + (reverb * reverbGain));

}).add;
)

In Parallel



In Series

• If we place the comb filters in series, we get a different effect.
• It only takes roughly four or five filters to achieve the desired echo 

density
• Each single impulse’s repetitions are repeated exponentially.

• Amplitude response is not flat and certain tones will be brought out 
of the reverb.  See the coding example to hear the issue.  This is not 
desirable.  The parallel version produces a better reverb even if it is 
less efficient.

Comb 1 Comb 2 Comb NInput Output



(
SynthDef(\seriesCombReverbStereo, {

arg in, out = 0, maxdelaytime = 0.2, decaytime = 2, inputGain = 0.5, reverbGain = 0.9;
var reverb, input;
input = In.ar(in, 2);
reverb = input;
// Successively smaller delays to attempt to mirror the late reflection times
[0.03, 0.0221, 0.0114, 0.006, 0.0041].do({

arg delaytime;
reverb = CombN.ar(reverb, maxdelaytime, delaytime, decaytime, delaytime * 30);

});
Out.ar(out, (input * inputGain) + (reverb * reverbGain));

}).add;
)

In Series

Not very good at all!



Schroeder Reverb

• In 1962, Manfred Schroeder published an important paper on several 
algorithms for natural sounding reverberation.
• He attempted to solve two problems

• Creating a reverb with a flat amplitude response.  A non-flat response leads to “coloration”.
• Achieving a suitable echo density of >1000 impulses per second.  Anything lower would lead 

to “flutter” and an unrealistic reverberation.
• Identified the need for an allpass delay (i.e., an allpass filter) that would have a flat 

amplitude response in IIR filter (i.e., one with feedback).
• Would prevent the issue of coloration

• His algorithm required a bank of parallel comb filters, as we have just 
discussed, as well as a series of allpass filters.
• Schroeder’s reverb has a more metallic sound than we desire today.



Allpass Filter

• The allpass filter is a combination of feedforward and feedback comb filter in series.
• The difference equation given an input 𝑥[𝑛], an output 𝑦[𝑛], and an intermediate step of 
𝑣[𝑛] can be summarized in two equations:

𝑣[𝑛]

𝑣[𝑛] = 𝑥[𝑛] − 𝑔 ∗ 𝑣[𝑛 − 𝑀]
𝑦[𝑛] = 𝑔 ∗ 𝑣[𝑛] + 𝑣[𝑛 − 𝑀]



Key Aspects of All Pass Filter

• Allpass filters are “colorless”.  The amplitude response of an allpass
filter is 1 for all frequencies so none are attenuated/amplified.
• Allpass filters do alter the phase response of the input signal.  

• Useful for phase rotation and phasers
• Useful for smearing transients

• Allpass filters are key components of artificial reverberation due to 
their colorless quality.



Schroeder Reverb

• Schroeder’s basic design for a 
natural sounding reverb was a 
bank of parallel comb filters to 
mimic the early reflections

• A series of all pass filters are 
used to add echo density and 
mimic the late reflections

• See his paper “Natural 
Sounding Artificial 
Reverberation” for more 
details about his specific 
implementation.  He used four 
parallel comb filters and two 
allpass filters and he makes 
specific markings on the 
settings for each.

Note how Schroeder mixes the 
original signal with 
reverberated signal.  This is 
common for nearly all reverbs.



Aside: Mix in SC

• The unit generator Mix can be used to mix down an array of signals 
into a single channel by adding the various signals together.  We will 
use this to create the parallel comb filter bank.
• Note that we could have used this UGen with our previous parallel comb filter 

bank.
• Particularly useful is the class method .fill

• 1st argument is the number of signals to mixdown into a single channel
• 2nd argument is a function expressing the signal that will be copied n times (n 

is given by the first argument)
• Most of the time we will employ randomization in the signal function for the 2nd

argument so that each individual signal is slightly different.  This allows us to create any n 
number of parallel comb filters but with randomized delay times for each, for example.



Schroeder Reverb in SC
(
SynthDef(\schroederReverbMonoInput, {

arg in, out = 0, reverbGain = 0.2;
var input, predelay, earlyRef, lateRef;

// Mono - can't modulate numChannels so would need separate UGen for stereo
input = In.ar(in, 1);

// Adding predelay can help create a more realistic reverberance 
predelay = DelayN.ar(input, 0.048);

// Early reflection comb filtering which 8 parallel comb filters
// Mix adds all the signals together
earlyRef = Mix.fill(7, {CombL.ar(predelay, 0.1, Rand(0.02, 0.08), 3)});

// Create a series of 4 AllPass Filters for late reflections
lateRef = earlyRef;
4.do({lateRef = AllpassN.ar(lateRef, 0.050, [0.050.rand, 0.050.rand], 0.7)}); // Now Stereo

Out.ar(out, input + (reverbGain * lateRef));
}).add;
)



”Eerier” reverb

• We can modify the Schroeder reverb on the previous slide, by 
modulating the delay of the parallel comb filter bank.
• The unit generator LFNoise and LFNoise1 (the latter is linearly 

interpolated) can generate random values between 0 and +1 
• Try switching Rand(0.02, 0.08) with LFNoise1.kr(0.1.rand, 
0.04, 0.05)
• This will create a sliding effect that can produce a spooky/unnatural 

reverb as parts of the signal receive changing delay times in the comb 
filtering stage.



Infinite Reverb

• The Schroeder Reverb attempts to create a natural sounding reverb.  
Let’s try and make something more unnatural but nevertheless 
sonically interesting.
• Consider a series of all pass filters that feedback upon themselves.

• Schroeder alludes to such a reverb in “Natural Sounding Artifical
Reverberation” but notes that such a system would produce a non-
exponential decay
• James A. Moorer’s ideal late reverberation was exponentially decaying white noise

• SuperCollider offers the UGens LocalIn and LocalOut to create a 
feedback loop within a SynthDef.
• Note: only one LocalIn/LocalOut may be used per SynthDef.



Infinite Reverb
SynthDef(\infiniteHoldReverb, {

arg feedback = 1, inputGain = 0.8, reverbGain = 0.3;
var sig, local;

// Input signal
sig = SoundIn.ar(0); // Mono input
local = LocalIn.ar(1) + sig;

// Apply Reverb
15.do{local = AllpassN.ar(local, 0.06, Rand(0.001, 0.06), 3)};

// Feedback loop
LocalOut.ar(local*feedback); // Consider adding a Low Pass Filter here

//Output
Out.ar(0, (inputGain * sig) + (reverbGain * local) ! 2); // Stereo Output

}).add;

Credit to Eli Fieldsteel



Infinite Reverb

• When the feedback gain is set to 1, the reverb is truly infinite and will 
continue to loop continuously.
• When the feedback <1, the reverb will decay.  Setting the feedback to 

something close to 1 will create a long decay and could be advantageous to 
continually add new sounds.  Keeping the reverb set at 1 could lead to 
clipping.

• When the feedback >1, the reverb will increase and gain and will lead to 
clipping and distortion.  You should not set it above one.

• One could consider adding a Low Pass Filter at the LocalOut stage.
• Most higher frequencies tend to be absorbed upon reflection and so digitally 

filtering out higher frequencies can lead to a more “realistic” sound.



Convolution Reverb

• Digital/algorithmic attempts at natural reverb are difficult to create.
• Many academic papers discussing techniques/strategies
• Many commercial digital reverbs are proprietary – don’t know how they are created

• One idea: attempt to capture the natural reverb of surrounding 
environment (i.e., concert hall, church, room… etc.)
• Impulse response of environment

• Impulse
• Sine Sweep

• Convolve the impulse response of the environment with a digital audio 
signal to create a natural sounding reverb.
• See the unit generator Convolution


